
GENERAL RULES FOR 8086/8088 INSTRUCTIONS

1.  An immediate value (i.e., a constant ) cannot be a destination operand.

For example, the following is invalid:
ADD 34BFh , BX

2.  The destination and source operands must be of the same size, except if the source operand is an immediate value of a lower size than the destination.
For example, the following is valid:
MOV AX , 03H
3.  Direct memory to memory transfer is invalid, except for string instructions. Such a transfer must be done indirectly using a general register of the appropriate size.
For example, if VAR1 and VAR2 are word variables (i.e., word memory operands) then the transfer:

MOV VAR1 , VAR2
is invalid. The transfer can be done as:
MOV  AX ,  VAR1
MOV  VAR2 ,  AX
4.  The transfer:
 MOV SegmentRegister , SegmentRegister
is invalid for all segment registers: CS, DS, SS, ES.
5.  The transfer:
 MOV SegmentRegister ,  ImmediateValue
is invalid for all Segment Registers: CS, DS, SS, ES.  Such a transfer can be done indirectly using a 16-bit general register.
For example, if a program has a data segment that is defined by a simplified segment directive .DATA, the DS  register can be initialized by: 
MOV  BX,  @DATA
MOV  DS,  BX
6.  The IP, CS, SS, and SP registers cannot be destination operands.

For example, the following is invalid:
MOV CS ,  2345
SOME 8086 INSTRUCTIONS

	INSTRUCTION
	EFFECT

	MOV Destination ,  Source
	Destination  :=  Source

	INC  Destination
	Destination  :=  Destination  +  1

(The operand is treated as an unsigned number)

	DEC  Destination
	Destination  :=  Destination  -  1

(The operand is treated as an unsigned number)

	ADD  Destination ,  Source
	Destination  :=  Destination  +  Source

(Both operands may have signed or unsigned numbers)

	SUB  Destination ,  Source
	Destination  :=  Destination  -  Source

(Both operands may have signed or unsigned numbers)

	XCHG  Destination ,  Source
	Exchanges the contents of the source and destination. Both Source and Destination cannot be Immediate operands.


The INTerrupt Instruction


INT   InterruptNumber

where InterruptNumber ranges from 00H to 0FFH (i.e., from 0 to 255). 

The execution of an INT instruction causes an Interrupt Service Routine (ISR) associated with the InterruptNumber to be executed. Many of the ISRs have multiple subfunctions. To specify which subfunction is to be executed under a particular InterruptNumber, the AH register is assigned a subfunction number before the execution of the INT instruction. Example:


MOV AH , 08H


INT 21H

causes subfunction number 08H of Interrupt number 21H to be executed. In addition, some subfunctions require other values to be passed to the ISR in particular registers. Example: Subfunction 09H of Interrupt 21H displays a $-terminated string on the screen. The subfunction requires the offset of that string to be passed in the DX register:


MOV DX , OFFSET STRING


MOV AH , 09H


INT 21H

DOS FUNCTION CALLS (INT 21H)

DOS function calls preserve the contents of all the registers except the AX register and any other register or registers in which they explicitly return data.

1. TERMINATE PROGRAM AND RETURN TO DOS (DOS FUNCTION 4CH)


 MOV  AH ,  4CH


 INT  21H

   
  A code of 00H in the AL register indicates normal program termination. Thus the function is usually invoked as:


MOV  AX ,  4C00H


INT  21H

2.  CHARACTER INPUT WITH ECHO (DOS FUNCTION CALL 01H)

MOV  AH ,  01H

INT  21H

The code of the input character is returned in the AL register.
3.  CHARACTER INPUT WITHOUT ECHO (DOS FUNCTION CALL 08H)

MOV  AH ,  08H

INT  21H

 The code of the input character is returned in the AL register.
4.  CHARACTER OUTPUT (DOS FUNCTION CALL 02H)

MOV  AH ,  02H

MOV  DL ,  Character   or   CharacterCode

INT  21H

  The first two lines may be interchanged.

Note: One way of moving the cursor to the beginning of the next output line is by outputting 0Dh (Carriage Return) followed by 0Ah (Line feed):

MOV  AH ,  02H

MOV  DL ,  0DH

INT  21H

MOV  DL ,  0AH

INT  21H

5.  DISPLAYING A $-TERMINATED STRING CONTAINING NO OTHER $-CHARACTER (DOS FUNCTION 09H)

   .DATA 

      STRING_NAME  DB  ‘THE STRING TO BE DISPLAYED’ , ‘$’


.CODE

      . . .

      MOV  AH ,  09H

                    MOV  DX ,  OFFSET  STRING_NAME

      INT  21H

      .  .  .
Note: 

· The terminating $ is not displayed, even if it appears within the string. Thus this function cannot be used to   display the $ character on the screen.

· The lines MOV  AH ,  09H  and MOV DX , OFFSET STRING_NAME may be interchanged.

· The OFFSET operator returns the offset address of a variable or label. Also the statement:

MOV  DX ,  OFFSET  STRING_NAME

 is equivalent to the statement:

         LEA  DX ,  STRING_NAME

 LEA stands for Load Effective Address.

· If the terminating $ is omitted, the operation displays characters in the memory, after the string, until it finds   a $ character, if any.

· To move the cursor to the beginning of the next output line, after displaying a string, put 0Dh and 0Ah after the string and before the terminating $. Example:

PROMPT  DB  ‘PLEASE, ENTER YOUR NAME: ’ , 0Dh , 0Ah , ‘$’

· Another way of moving the cursor to the beginning of the next output line is to display , using DOS function 09H, a string of the form:

STRING1  DB  0Dh , 0Ah , ‘$’

6.  DISPLAYING A STRING WHICH MAY CONTAIN A $-CHARACTER (DOS FUNCTION 40H)

.DATA 

. . .
STRING_NAME  DB  ‘THE STRING TO BE DISPLAYED’

STRINGLEN  EQU   $  -  STRING_NAME 

. . .
.CODE

. . .
MOV  AH ,  40H

MOV  BX ,  01H





; file handle for the screen

MOV  CX ,  STRINGLEN



; string length

MOV  DX ,  OFFSET  STRING_NAME

INT  21H

. . .

· The EQU directive defines a value that the assembler can use to substitute in other instructions.

· MASM or TASM makes two passes through the source file. On the first pass, MASM or TASM checks for syntax errors and creates a symbol table of names and their relative locations within a segment. To keep track of locations, it uses a location counter. The location counter is reset to 0 at the beginning of a segment. When an instruction is encountered, the location counter is increased by the number of bytes needed for the machine code of the instruction. When a name is encountered, it is entered in the symbol table along with the location counter’s value.
· An operand containing a dollar symbol, $, refers to the current value in the location counter. Thus, in the above example $ - STRING_NAME evaluates to the number of bytes between STRING_NAME and STRINGLEN which is the number of bytes (i.e., characters) in ‘THE STRING TO BE DISPLAYED’

7. BUFFERED KEYBOARD INPUT (READING A STRING)(DOS FUNCTION 0AH)

(a) To read a string, a buffer (i.e., an array) must be defined to store that string. One way of defining the buffer is:

BUFFER_NAME  DB  Num1 , Num2 DUP(?)

where Num1 is the maximum number of string characters to be read including the Carriage Return (0Dh), which is also read and stored in the buffer as the last string character. Num2 has a value which is one more than Num1; it is the number of bytes reserved to hold the actual length of the string (i.e., minus the Carriage Return character), the string, and the terminating Carriage Return character. Note: The maximum value for Num1 is FEh i.e., 254

Example: Define a buffer called BUFFER to store a string of maximum length 50:

Solution:

BUFFER  DB  51, 52  DUP(?)

      stored in this byte 

(b) To read a string from the keyboard into a buffer called BUFFER as defined above, we invoke DOS function 0AH as:

MOV  AH ,  0AH
MOV  DX ,  OFFSET  BUFFER
INT 21H
The operation echoes the entered characters on the screen and advances the cursor. If more characters than the specified maximum (in byte 0) are entered, the speaker beeps and the additional characters are not read.
(c) Displaying a buffer, defined in 7(a) by using DOS function 40H:


MOV  AH ,  40H



MOV  BX ,  01H


; file handle for the screen



MOV  CH , 00H


; Initialize CX with the string length



MOV  CL ,  BUFFER[1]

;



LEA DX , BUFFER[2]

; Output the string starting at byte 2



INT  21H


;

 Note:  The statement :



MOV  DX ,  OFFSET  BUFFER[2]


    is equivalent to:



LEA  DX ,  BUFFER[2]

 (d)  BUFFERED KEYBOARD INPUT (READING A STRING)(DOS FUNCTION 0AH): 

       ANOTHER WAY OF DEFINING THE BUFFER

To read a string of , say, maximum length 50 the buffer can be defined as:




MAXLEN  DB   51




ACTLEN   DB    ?




BUFFER    DB    51  DUP(?)


To read a string into the buffer, we invoke DOS function 0AH as:




MOV   AH ,  0AH




MOV   DX ,  OFFSET  MAXLEN




INT   21H

(e) Displaying a buffer, defined in 7(d) by using DOS function 40H:


MOV  AH ,  40H



MOV  BX ,  01H


; file handle for the screen



MOV  CH , 00H


; Initialize CX with the string length



MOV  CL ,  ACTLEN

;



LEA    DX , BUFFER

; Display the buffer



INT  21H


;
